Building Trading Systems On NonLinear Filters

Presented by John Ehlers

Disclaimer

Each speaker at the TradeStationWorld Conference acts independently, and no speaking topic, session, seminar or content is affiliated with, or approved, sponsored or endorsed by, TradeStation Technologies, Inc. or any of its affiliates. Topics, sessions and seminars are solely for educational purposes. The speaker roster and session/seminar content are subject to change without notice. No investment or trading advice regarding any security, group of securities, market segment or market is intended or shall be given. Any examples used in sessions, seminars or speaking topics are for illustrative purposes only -- they should never be construed as recommendations or endorsements of any kind. No particular trading strategy, technique, method or approach discussed will guarantee profits, increased profits or the minimization of losses. Past performance, whether actual or indicated by simulated historical tests, is no guarantee of future performance or success.

Testimonials may not be representative of the experiences of other customers and are not indicative of future performance or success.

TradeStation Technologies, Inc., the host of the conference, and TradeStation Securities, Inc. (Member NASD, SIPC and NFA), the conference's premier sponsor, are affiliated companies. "TradeStation," as used in this presentation, refers to the trading analysis software products, platforms and services that have been developed by TradeStation Technologies.

System Design Approach

- Pick nonlinear moving averages as the base technology to react to major moves and avoid whipsaws in sideways markets

- Survey available nonlinear moving averages
- Select a time-based event
- Include volatility guards
- Add loss escape mechanism

Nonlinear Moving Average Survey

- Eight Approaches are considered
  - KAMA (Kaufman Adaptive Moving Average)
  - VIDYA (Variable Index Dynamic Average)
  - MAMA (MESA Adaptive Moving Average)
  - Ehlers Filter
  - Median Filter
  - Median-MA Difference Filter
  - FRAMA (Fractal Adaptive Moving Average)
  - Nonlinear Laguerre Filter

KAMA (Kaufman Adaptive Moving Average)

- Describe by Perry Kaufman in “Trading Systems and Methods”, Third Edition, John Wiley & Sons, pp 436-438
- Adjusts the alpha of an EMA according to volatility
  - Ratios the price difference over a time span to the sum of the bar-to-bar price differences over the time span
- Alpha is limited to range between a lower and upper bound

KAMA Indicator EL Code

```
Inputs:    Price((H+L)/2), Len(10),FastLen(2), SlowLen(30);

Vars:      count(0), Num(0), Denom(0),ER(0), Fastest(0), Slowest(0),
alpha(0), Filt(0);

Num = AbsValue(Price - Price[Len]);
Denom = 0;
For count = 0 to Len begin
Denom = Denom + AbsValue(Price[count] - Price[count + 1]);
End;
If Denom <> 0 then ER = Num / Denom;
Fastest = 2 / (FastLen + 1);
Slowest = 2 / (SlowLen + 1);
alpha = Square(ER*(Fastest - Slowest) + Slowest);
Filt = alpha*Price + (1 - alpha)*Filt[1];
If CurrentBar = 1 then Filt = Price;

Plot1(Filt, "KAMA");
```

VIDYA (Variable Index Dynamic Average)

- Developed by Tushar Chande and Stanley Kroll in “The new Technical Trader”, John Wiley & Sons, 1984
- Dynamically adjusts the alpha of an EMA according the the ratio of the Standard Deviation of prices over a period to the Standard Deviation of prices over a longer period
  - Modifies the alpha of a suggested 9 bar EMA
    - alpha = 2 / (length + 1)

VIDYA Indicator EL Code

```
Inputs:    M(30), N(9);

Vars:      k(0), VIDYA(0);

If StdDev(Close, M) <> 0 then k = StdDev(Close, N) / StdDev(Close, M);

Filt = .2*k*Close + (1 - .2*k)*Filt[1];
If CurrentBar = 1 then Filt = Close;

Plot1(Filt, "VIDYA");
```

MAMA (MESA Adaptive Moving Average)

- Uses the Hilbert Transform to measure the current dominant cycle period
- Computes the phase of the dominant cycle
- Computes the alpha inversely proportional to the rate change of phase
  - Shorter periods have higher rate changes of phase
    - Large alpha - more responsive EMA
  - Longer periods have lower rate changes of phase
    - Smaller alpha - gives EMA more smoothing
- High rate change of phase due to snap-back every 180 degrees ensures EMA will tightly follow price

MAMA Indicator EL Code

```
Inputs:          Price((H+L)/2),     speed(.8), FastLimit(.5), SlowLimit(.05);
Vars:            Smooth(0), Detrender(0), I1(0), Q1(0), jI(0), jQ(0), I2(0), Q2(0), Re(0), Im(0), Period(0), SmoothPeriod(0), Phase(0),
DeltaPhase(0), alpha(0), Filt(0);
If CurrentBar > 5 then begin
Smooth = (4*Price + 3*Price[1] + 2*Price[2] + Price[3]) / 10;
Detrender = (.0962*Smooth + .5769*Smooth[2] - .5769*Smooth[4] - .0962*Smooth[6])*(.075*Period[1] + .54);
{Compute InPhase and Quadrature components}
Q1 = (.0962*Detrender + .5769*Detrender[2] - .5769*Detrender[4] - .0962*Detrender[6])*(.075*Period[1] + .54);
                I1 = Detrender[3];
{Advance the phase of I1 and Q1 by 90 degrees}
jI = (.0962*I1 + .5769*I1[2] - .5769*I1[4] - .0962*I1[6])*(.075*Period[1] + .54);
jQ = (.0962*Q1 + .5769*Q1[2] - .5769*Q1[4] - .0962*Q1[6])*(.075*Period[1] + .54);
{Phasor addition for 3 bar averaging)}
I2 = I1 - jQ;
Q2 = Q1 + jI;
{Smooth the I and Q components before applying the discriminator}
I2 = .2*I2 + .8*I2[1];
Q2 = .2*Q2 + .8*Q2[1];
{Homodyne Discriminator}
Re = I2*I2[1] + Q2*Q2[1];
Im = I2*Q2[1] - Q2*I2[1];
Re = .2*Re + .8*Re[1];
Im = .2*Im + .8*Im[1];
If Im <> 0 and Re <> 0 then Period = 360/ArcTangent(Im/Re);
If Period > 1.5*Period[1] then Period = 1.5*Period[1];
If Period < .67*Period[1] then Period = .67*Period[1];
If Period < 6 then Period = 6;
If Period > 50 then Period = 50;
Period = .2*Period + .8*Period[1];
SmoothPeriod = .33*Period + .67*SmoothPeriod[1];
If I1 <> 0 then Phase = (ArcTangent(Q1 / I1));
DeltaPhase = Phase[1] - Phase;
If DeltaPhase < 1 then DeltaPhase = 1;
alpha = Speed / DeltaPhase;
If alpha < SlowLimit then alpha = SlowLimit;
If alpha > FastLimit then alpha = FastLimit;
Filt = alpha*Price + (1 - alpha)*Filt[1];
End;
If CurrentBar <= 5 then Filt= Price;
Plot1(Filt, "MAMA");
```

Ehlers Filter

- Unlike most nonlinear filters, it is a FIR filter
- Analogous to determining sharpness of a piece of paper creased and draped over the edge of a table



- FIR coefficients are computed as “distance” vectors - squared price differences are summed
- Coefficients are normalized to their sum for unity gain

Ehlers Filter EL Code

```
Inputs:      Price((H+L)/2), Length(20);

Vars:        Smooth(0), count(0), LookBack(0), SumCoef(0), Num(0), Filt(0);

Array:       Coef[50](0), Distance2[50](0);

Smooth = (Price + 2*Price[1] + 2*Price[2] + Price[3]) / 6;
For count = 0 to Length -1 begin
   Distance2[count] = 0;
   For Lookback = 1 to Length-1 begin
      Distance2[count] = Distance2[count] + (Smooth[count] - Smooth[count + Lookback])*(Smooth[count]
- Smooth[count + Lookback]);
   End;
   Coef[count] = Distance2[count];
End;
Num = 0;
SumCoef = 0;
For count = 0 to Length -1 begin
   Num = Num + Coef[count]*Smooth[count];
   SumCoef = Sumcoef + Coef[count];
End;
If SumCoef <> 0 then Filt = Num / SumCoef;

Plot1(Filt, "Ehlers");
```

Median Filter

- Rank-order filter
- Easy to compute
- Often used to sharpen video images
- Useful to smooth impulsive type noise by ignoring outliers

Median Filter EL Code

```
Inputs: Price((H+L)/2), Len(4);

Vars:      Filt(0);

Filt = Median(Price, 2*Len + 1);

Plot1(Filt, "Median");
```

Median-MA Difference Filter

- Adjusts the alpha of an EMA according to the differential responses of Median and MA filters
- Consider a price string of ten 1s
  - Both the Median and MA is 1
- New price data point has a value of 10
  - Median output is still 1 (new price value is ignored)
  - Simple MA value is 1.9
- Searches for a filter length where the output differences fall below a selected threshold
  - Fast moving markets produce the shortest (most responsive) filter

Median-MA Difference Filter EL Code

```
Inputs:     Price((H+L)/2), Threshold(.0025);

Vars:       Smooth(0), Length(30), alpha(0), Filt(0);

Smooth = (Price + 2*Price[1] + 2*Price[2] + Price[3]) / 6;

Length = 39;
Value3 = 1;
While Value3 > Threshold begin
alpha = 2 /(Length + 1);
Value1 = Median(Smooth, Length);
Value2 = alpha*Smooth + (1 - alpha)*Value2[1];
If Value1 <> 0 then Value3 = AbsValue(Value1 - Value2) / Value1;
Length = Length - 2;
End;
If Length < 3 then Length = 3;
alpha = 2 /(Length + 1);

Filt = alpha*Smooth + (1 - alpha)*Filt[1];
If CurrentBar < 4 then Filt = Price;

Plot1(Filt, "Med-MA");
```

FRAMA (Fractal Adaptive Moving Average)

- There is no argument that the market moves as a fractal
- A period is selected to compute the fractal dimension
  - The price difference over the first half of the range, second half of the range, and over the total range is used for the computation
- Since the market prices move as log-normal, the fractal dimension is used to compute filter alpha as

$$\alpha = \exp(-4.6 * (Dimen - 1))$$

  - When Dimen = 1, α = 1 - a very fast filter
  - When Dimen = 2, α = .01 - about a 200 bar filter

FRAMA Filter EL Code

```
Inputs:        Price((H+L)/2), N(20);

Vars:          count(0), N1(0), N2(0), N3(0), HH(0), LL(0), Dimen(0), alpha(0), Filt(0);

N3 = (Highest(High, N) - Lowest(Low, N)) / N;
HH = High;
LL = Low;
For count = 0 to N/2 - 1 begin
If High[count] > HH then HH = High[count];
If Low[count] < LL then LL = Low[count];
End;
N1 = (HH - LL)/(N/2);
HH = High[N/2];
LL = Low[N/2];
For count = N/2 to N - 1 begin
If High[count] > HH then HH = High[count];
If Low[count] < LL then LL = Low[count];
End;
N2 = (HH - LL)/(N/2);

If N1 > 0 and  N2 > 0 and N3 > 0 then Dimen = (Log(N1 + N2) - Log(N3)) / Log(2);

{alpha = .02 when Dimen = .7 and alpha = .33 when Dimen = .05}
alpha = ExpValue(-4.6*(Dimen - 1));
If alpha < .01 then alpha = .01;
If alpha > 1 then alpha = 1;

Filt = alpha*Price + (1 - alpha)*Filt[1];
If CurrentBar < N + 1 then Filt = Price;

Plot1(Filt, "FRAMA");
```

NonLinear Laguerre

- A Laguerre filter warps time in the filter coefficients
  - Enables extreme smoothing with just a few filter terms
- A NonLinear Laguerre filter measures the difference between the current price and the last computed filter output.
  - Objective is to drive this “error” to zero
  - The “error”, normalized to the error range over a selected period is the alpha of the Laguerre filter

Nonlinear Laguerre Filter EL Code

```
Inputs:     Price((H+L)/2), Length(20);
Vars:       Diff(0), HH(0), LL(0), count(0), alpha(0), L0(0), L1(0), L2(0), L3(0), Filt(0), FIR(0);

Diff = AbsValue(Price - Filt[1]);
HH = Diff;
LL = Diff;
For count = 0 to Length - 1 begin
If Diff[count] > HH then HH = Diff[count];
If Diff[count] < LL then LL = Diff[count];
End;
If CurrentBar > Length and HH - LL <> 0 then alpha = Median(((Diff - LL) / (HH - LL)), 5);
L0 = alpha*Price + (1 - alpha)*L0[1];
L1 = -(1 - alpha)*L0 + L0[1] + (1 - alpha)*L1[1];
L2 = -(1 - alpha)*L1 + L1[1] + (1 - alpha)*L2[1];
L3 = -(1 - alpha)*L2 + L2[1] + (1 - alpha)*L3[1];
Filt = (L0 + 2*L1 + 2*L2 + L3) / 6;
If CurrentBar < Length then begin
L0 = Price;
L1 = Price;
L2 = Price;
L3 = Price;
Filt = Price;
End;

Plot1(Filt, "Laguerre");
```

NonLinear Filter Comparison (1)



NonLinear Filter Comparison (2)



Filter Selection Process

- Rank each filter according to smoothness on a scale from 1 to 8
- Rank each filter according to responsiveness on a scale from 1 to 8
- Add the rankings to obtain a score
  - low score is the best filter for the job

	Smoothness	Responsive	Score
KAMA	5	6	11
VIDYA	6	5	11
MAMA	7	1	8
Ehlers	1	3	4
Median	8	7	15
Median-MA	4	8	12
FRAMA	2	4	6
Laguerre	3	2	5

Time Based Event

- Line crossings are distinctive events easily identified in automatic systems
- Create a trigger by delaying the nonlinear filter by one bar

- The problem is that crossings create whipsaws in sideways markets - need hysteresis

Volatility-Based Histeresis Channel

- Measure the average range (can use Average True Range if desired)
- Add and subtract a fraction of the average range to the NonLinear Filter

Trading Strategy Code

- Any NonLinear Filter can be used
- To NonLinear Filter code add:
  - Also declare Rng variable and add Frac Input

```
Rng = .1*(High - Low) + .9*Rng[1];
Value1 = Filt[1] + Rng / Frac;
Value2 = Filt[1] - Rng / Frac;

If Filt Crosses Over Value1 Then Buy Next Bar on Open;
If Filt Crosses Under Value2 Then Sell Short Next Bar on Open;
```

- Trading Rules are simple
  - Contains time event trigger as Filt[1]
  - Contains Hysteresis channel as ± Rng/Frac
- Trading system is always in the market - reversing between long and short positions
  - Excellent approach for Commodities and ETFs
  - Long-Only positions for stocks can be taken

Efficiency Tips

- Precede indicator or strategy name with a special character like “!” or “*” or “=“
  - This moves your custom indicators and strategies to the top of the TradeStation list
- Precede research indicators and strategies with double special characters like “!!” or “**” or “==“
  - Avoids versionitis - you know you can delete one of these later without worrying about losing content
  - Easy converts to a custom indicator or strategy simply by removing one of the special characters

Optimization Tips

- Optimize one parameter at a time for efficiency
  - Iterate if necessary
- A good optimization will have a gentle “mound”
- Parameters should be robust over a wide range





Maximum Adverse Excursion

- Large losers indicate a loss escape is desirable



- Add the following code to the system to reverse from a losing position

```
If MarketPosition = 1 then Sell Short at EntryPrice - PtStop Stop;
If MarketPosition = -1 then Buy at EntryPrice + PtStop Stop;
```

Complete System EL Code

```
Inputs:       Price((H+L)/2), Length(20), Frac(5), PtStop(3);
Vars:         Smooth(0), count(0), LookBack(0), SumCoef(0), Num(0), Filt(0), Rng(0);
Array:        Coef[50](0), Distance2[50](0);
Smooth = (Price + 2*Price[1] + 2*Price[2] + Price[3]) / 6;
For count = 0 to Length -1 begin
   Distance2[count] = 0;
   For Lookback = 1 to Length-1 begin
      Distance2[count] = Distance2[count] + (Smooth[count] - Smooth[count + Lookback])*(Smooth[count]
- Smooth[count + Lookback]);
   End;
   Coef[count] = Distance2[count];
End;
Num = 0;
SumCoef = 0;
For count = 0 to Length -1 begin
   Num = Num + Coef[count]*Smooth[count];
   SumCoef = Sumcoef + Coef[count];
End;
If SumCoef <> 0 then Filt = Num / SumCoef;
Rng = .1*(High - Low) + .9*Rng[1];
Value1 = Filt[1] + Rng / Frac;
Value2 = Filt[1] - Rng / Frac;
If Filt Crosses Over Value1 Then Buy Next Bar on Open;
If Filt Crosses Under Value2 Then Sell Short Next Bar on Open;
If MarketPosition = 1 then Sell Short at EntryPrice - PtStop Stop;
If MarketPosition = -1 then Buy at EntryPrice + PtStop Stop;
```

System Development Tip

- Plot Open and Closed Equity to identify problematic trades

INDICATOR

```
Plot1(I_ClosedEquity, "Closed");
Plot2(I_OpenEquity, "Open");
```

29 Year SF Equity Growth



24 Year JY Equity Growth



Six Year EC Equity Growth



29 Year SF Performance

TradeStation Strategy Performance Report - !2005TSWorld1 @SF.CSV-Daily

Performance Summary: All Trades

Total Net Profit	$163,887.50	Open position P/L	$200.00
Gross Profit	$354,862.50	Gross Loss	($190,975.00)
Total # of trades	222	Percent profitable	50.45%
Number winning trades	112	Number losing trades	110
Largest winning trade	$17,962.50	Largest losing trade	($5,962.50)
Average winning trade	$3,168.42	Average losing trade	($1,736.14)
Ratio avg win/avg loss	1.82	Avg trade (win & loss)	$738.23
Max consec. Winners	5	Max consec. losers	5
Avg # bars in winners	48	Avg # bars in losers	17
Max intraday drawdown	($15,150.00)		
Profit Factor	1.86	Max # contracts held	1
Account size required	$15,150.00	Return on account	1081.77%

24 Year JY Performance

TradeStation Strategy Performance Report - !2005TSWorld1 @JY.CSV-Daily

Performance Summary: All Trades

Total Net Profit	$166,887.50	Open position P/L	$1,762.50
Gross Profit	$393,287.50	Gross Loss	($226,400.00)
Total # of trades	313	Percent profitable	38.34%
Number winning trades	120	Number losing trades	193
Largest winning trade	$24,475.00	Largest losing trade	($2,625.00)
Average winning trade	$3,277.40	Average losing trade	($1,173.06)
Ratio avg win/avg loss	2.79	Avg trade (win & loss)	$533.19
Max consec. Winners	5	Max consec. losers	7
Avg # bars in winners	36	Avg # bars in losers	8
Max intraday drawdown	($14,925.00)		
Profit Factor	1.74	Max # contracts held	1
Account size required	$14,925.00	Return on account	1118.17%

6 Year EC Performance

TradeStation Strategy Performance Report - !2005TSWorld1 @EC.CSV-Daily (1/4/1999-3/4/2005)

Performance Summary: All Trades

Total Net Profit	$79,825.00	Open position P/L	$875.00
Gross Profit	$122,062.50	Gross Loss	($42,237.50)
Total # of trades	50	Percent profitable	48.00%
Number winning trades	24	Number losing trades	26
Largest winning trade	$13,800.00	Largest losing trade	($4,875.00)
Average winning trade	$5,085.94	Average losing trade	($1,624.52)
Ratio avg win/avg loss	3.13	Avg trade (win & loss)	$1,596.50
Max consec. Winners	3	Max consec. losers	2
Avg # bars in winners	45	Avg # bars in losers	16
Max intraday drawdown	($7,875.00)		
Profit Factor	2.89	Max # contracts held	1
Account size required	$7,875.00	Return on account	1013.65%

The System Works on ETFs

TradeStation Strategy Performance Report - !2005TSWorld1 QQQQ.TXT-Daily (3/10/1999-5/13/2005

Performance Summary: All Trades

Total Net Profit	$8,565.00	Open position P/L	$232.00
Gross Profit	$14,266.00	Gross Loss	($5,701.00)
Total # of trades	33	Percent profitable	36.36%
Number winning trades	12	Number losing trades	21
Largest winning trade	$4,940.00	Largest losing trade	($413.00)
Average winning trade	$1,188.83	Average losing trade	($271.48)
Ratio avg win/avg loss	4.38	Avg trade (win & loss)	$259.55
Max consec. Winners	3	Max consec. losers	6
Avg # bars in winners	86	Avg # bars in losers	18
Max intraday drawdown	($2,288.00)		
Profit Factor	2.50	Max # contracts held	1
Account size required	$2,288.00	Return on account	374.34%



Object Lesson:
You can't make money unless the market moves

The System Works on Stocks, Too

		CSCO			RTN	
	All Trades	Long Trades	Short Trades	All Trades	Long Trades	Short Trades
Total Net Profit	$8,348.00	$5,093.00	$3,255.00	$3,980.00	$2,194.00	$1,786.00
Gross Profit	$12,873.00	$7,775.00	$5,098.00	$4,366.00	$2,510.00	$1,856.00
Gross Loss	($4,525.00)	($2,682.00)	($1,843.00)	($386.00)	($316.00)	($70.00)
Profit Factor	2.84	2.9	2.77	11.31	7.94	26.51
Open Position P/L	$42.00	$42.00	$0.00	($243.00)	$0.00	($243.00)
Total Number of Trades	72	36	36	15	8	7
Percent Profitable	45.83%	58.33%	33.33%	66.67%	62.50%	71.43%
Winning Trades	33	21	12	10	5	5
Losing Trades	37	15	22	4	2	2
Even Trades	2	0	2	1	1	0
Avg. Trade Net Profit	$115.94	$141.47	$90.42	$265.33	$274.25	$255.14
Avg. Winning Trade	$390.09	$370.24	$424.83	$436.60	$502.00	$371.20
Avg. Losing Trade	($122.30)	($178.80)	($83.77)	($96.50)	($158.00)	($35.00)
Ratio Avg. Win:Avg. Loss	3.19	2.07	5.07	4.52	3.18	10.61
Largest Winning Trade	$4,320.00	$4,179.00	$4,320.00	$1,320.00	$1,320.00	$944.00
Largest Losing Trade	($437.00)	($437.00)	($241.00)	($215.00)	($215.00)	($35.00)
Max. Consecutive Winni	3	9	3	4	2	4
Max. Consecutive Losing	6	3	6	2	1	1
Avg. Bars in Total Trades	53.56	72.83	34.28	56.07	72.88	36.86
Avg. Bars in Winning Tra	87.73	105.1	57.33	71.7	103	40.4
Avg. Bars in Losing Trad	25.16	27.67	23.45	22.75	17.5	28
Avg. Bars in Even Trades	15	0	15	33	33	0
Trading Period	15 Yrs, 10 Dys			3 Yrs, 9 Mths, 17 I		
Max. Equity Run-up	$10,025.00			$4,661.00		
Date of Max. Equity Run	4/4/2001 13:00			12/1/2004 13:00		
Max. Drawdown (Intra-day Peak to Valley)						
Value	($2,694.00)	($3,027.00)	($1,859.00)	($817.00)	($699.00)	($685.00)
Date	4/14/2000 13:00			5/9/2005 13:00		
Max. Drawdown (Trade Close to Trade Close)						
Value	($1,633.00)	($1,558.00)	($733.00)	($250.00)	($215.00)	($35.00)
Date	2/7/2003 13:00			9/24/2003 13:00		

SUMMARY

- Eight (or more) systems can be created using NonLinear Moving Averages as a basis
- The systems have four components
  - NonLinear Moving Average
  - Time Event
  - Volatility Histeresis
  - Loss Escape Mechanism
- The systems are robust over long time spans
- The systems are robust over vastly different trading instruments